

1.1 Introduction to package `qm`

The `qm` package was written by Eric Majzoub, University of Missouri. Email: majzoub@at-umsystem.edu

The `qm` package provides functions and standard definitions to solve quantum mechanics problems in a finite dimensional Hilbert space. For example, one can calculate the outcome of Stern-Gerlach experiments using the built-in definition of the S_x , S_y , and S_z operators for arbitrary spin, e.g. $s=\{1/2, 1, 3/2, \dots\}$. For spin-1/2 the standard basis states in the x , y , and z -basis are available as $\{x_p, x_m\}$, $\{y_p, y_m\}$, and $\{z_p, z_m\}$, respectively. One can create general ket vectors with arbitrary but finite dimension and perform standard computations such as expectation value, variance, etc. The angular momentum $|j, m\rangle$ representation of kets is also available. It is also possible to create tensor product states for multiparticle systems and to perform calculations on those systems.

Let us consider a simple example involving spin-1/2 particles. A bra vector in the z -basis may be written as

$$\langle \psi | = a \langle z+ | + b \langle z- |.$$

The bra will be represented in Maxima by the row vector $[a \ b]$, where the basis vectors are

$$\langle z+ | = [1 \ 0]$$

and

$$\langle z- | = [0 \ 1].$$

There are two types of kets and bras available in this package, the first type is given by a *matrix representation*, as in the above example. **kets** are column vectors and **bras** are row vectors, and their components are entered as Maxima *lists* in the **bra** and **ket** functions. The second type of bra or ket is *abstract*; there is no matrix representation. Abstract bras and kets are entered *without* using lists for the elements. Thus, if one wishes to do purely symbolic calculations, then input of abstract kets, (j, m) -kets, and so forth should be done without lists. If one wishes to do numerical or component computations using the kets then enter the arguments as a list. Note that abstract kets and bras are *assumed to be orthonormal*.

The following examples illustrate some of the basic capabilities of the `qm` package. Here both abstract, and concrete kets are shown.

```
(%i1) ket(a,b)+ket(c,d);
(%o1)                                     |c, d> + |a, b>
(%i2) ket([a,b,c])+ket([d,e,f]);
(%o2)                                     [ d + a ]
                                     [          ]
                                     [ e + b ]
                                     [          ]
                                     [ f + c ]
```

Next, tensor products of the spin-1/2 basis states $\{z_p, z_m\}$ are shown in abstract and matrix representations.

```
(%i1) ketprod('zp','zm')+ketprod('zm','zp);
(%o1) ketprod(zp, zm) + ketprod(zm, zp)
(%i2) ketprod([zp,zm]);
(%o2) [ 1 ] [ 0 ]
      [tpket, [[ ], [ ]]]
      [ 0 ] [ 1 ]
```

Abstract kets and bras are assumed to be orthonormal as shown in the following examples.

```
(%i1) declare([a,b],complex);
(%o1) done
(%i2) psi:a*ket(1)+b*ket(2);
(%o2) |2> b + |1> a
(%i3) psidag:dagger(psi);
(%o3) <2| conjugate(b) + <1| conjugate(a)
(%i4) psidag . psi;
(%o4) b conjugate(b) + a conjugate(a)
(%i1) declare([c1,c2],complex,r,real);
(%o1) done
(%i2) k:ket([c1,c2,r]);
(%o2) [ c1 ]
      [   ]
      [ c2 ]
      [   ]
      [ r  ]
(%i3) b:dagger(k);
(%o3) [ conjugate(c1) conjugate(c2) r ]
(%i4) b . k;
(%o4) r^2 + c2 conjugate(c2) + c1 conjugate(c1)
```

The package is loaded with: `load(qm);`

1.2 Functions and Variables for qm

hbar [Variable]
Planck's constant divided by 2π . `hbar` is not given a floating point value, but is declared to be a real number greater than zero.

ket ($[c_1, c_2, \dots]$) [Function]
ket creates a *column* vector of arbitrary finite dimension. The entries c_i can be any Maxima expression. The user must **declare** any relevant constants to be complex. For a matrix representation the elements must be entered as a list in $[\dots]$ square brackets. If no list is entered the ket is represented as a general ket, **ket(a)** will return $|a\rangle$.

```

(%i1) kill(a);
(%o1)                                     done
(%i2) ket(a);
(%o2)                                     |a>
(%i3) declare([c1,c2],complex);
(%o3)                                     done
(%i4) ket([c1,c2]);
(%o4)                                     [ c1 ]
                                     [   ]
                                     [ c2 ]
(%i5) facts();
(%o5) [kind(hbar, real), hbar > 0, kind(c1, complex), kind(c2, complex)]

```

bra ($[c_1, c_2, \dots]$) [Function]
bra creates a *row* vector of arbitrary finite dimension. The entries c_i can be any Maxima expression. The user must **declare** any relevant constants to be complex. For a matrix representation the elements must be entered as a list in $[\dots]$ square bras. If no list is entered the bra is represented as a general bra, **bra(a)** will return $\langle a|$.

```

(%i1) kill(c1,c2);
(%o1)                                     done
(%i2) bra(c1,c2);
(%o2)                                     <c1, c2|
(%i3) bra([c1,c2]);
(%o3)                                     [ c1  c2 ]
(%i4) facts();
(%o4)                                     [kind(hbar, real), hbar > 0]

```

ketp (*vector*) [Function]
ketp is a predicate function that checks if its input is a ket, in which case it returns **true**, else it returns **false**. **ketp** only returns **true** for the matrix representation of a ket.

```

(%i1) kill(a,b,k);
(%o1)                                     done
(%i2) k:ket(a,b);
(%o2)                                     |a, b>
(%i3) ketp(k);
(%o3)                                     false
(%i4) k:ket([a,b]);
(%o4)                                     [ a ]
                                     [   ]
                                     [ b ]
(%i5) ketp(k);
(%o5)                                     true

```

brap (*vector*) [Function]

brap is a predicate function that checks if its input is a bra, in which case it returns **true**, else it returns **false**. **brap** only returns **true** for the matrix representation of a bra.

```
(%i1) b:bra([a,b]);
(%o1)          [ a  b ]
(%i2) brap(b);
(%o2)          true
```

Two additional functions are provided to create kets and bras in the matrix representation. Additionally these functions attempt to automatically **declare** constants as complex. For example, if a list entry is **a*sin(x)+b*cos(x)** then only **a** and **b** will be **declare-d** complex and not **x**.

autoket ($[a_1, a_2, \dots]$) [Function]

autoket takes a list $[a_1, a_2, \dots]$ and returns a ket with the coefficients a_i **declare-d** complex. Simple expressions such as **a*sin(x)+b*cos(x)** are allowed and will **declare** only the coefficients as complex.

```
(%i1) autoket([a,b]);
(%o1)          [ a ]
              [  ]
              [ b ]

(%i2) facts();
(%o2) [kind(hbar, real), hbar > 0, kind(a, complex), kind(b, complex)]
(%i1) autoket([a*sin(x),b*sin(x)]);
(%o1)          [ a sin(x) ]
              [          ]
              [ b sin(x) ]

(%i2) facts();
(%o2) [kind(hbar, real), hbar > 0, kind(a, complex), kind(b, complex)]
```

autobra ($[a_1, a_2, \dots]$) [Function]

autobra takes a list $[a_1, a_2, \dots]$ and returns a bra with the coefficients a_i **declare-d** complex. Simple expressions such as **a*sin(x)+b*cos(x)** are allowed and will **declare** only the coefficients as complex.

```
(%i1) autobra([a,b]);
(%o1)          [ a  b ]

(%i2) facts();
(%o2) [kind(hbar, real), hbar > 0, kind(a, complex), kind(b, complex)]
(%i1) autobra([a*sin(x),b]);
(%o1)          [ a sin(x)  b ]

(%i2) facts();
(%o2) [kind(hbar, real), hbar > 0, kind(a, complex), kind(b, complex)]
```

dagger (*vector*) [Function]

dagger is the quantum mechanical *dagger* function and returns the **conjugate transpose** of its input.

```
(%i1) dagger(bra([%i,2]));
(%o1)          [ - %i ]
              [      ]
              [  2   ]
```

braket (psi,phi) [Function]

Given two kets **psi** and **phi**, **braket** returns the quantum mechanical bracket $\langle \text{psi} | \text{phi} \rangle$. The vector **psi** may be input as either a **ket** or **bra**. If it is a **ket** it will be turned into a **bra** with the **dagger** function before the inner product is taken. The vector **phi** must always be a **ket**.

```
(%i1) declare([a,b,c],complex);
(%o1)                                     done
(%i2) braket(ket([a,b,c]),ket([a,b,c]));
(%o2)          c conjugate(c) + b conjugate(b) + a conjugate(a)
```

norm (psi) [Function]

Given a **ket** or **bra** **psi**, **norm** returns the square root of the quantum mechanical bracket $\langle \text{psi} | \text{psi} \rangle$. The vector **psi** must always be a **ket**, otherwise the function will return **false**.

```
(%i1) declare([a,b,c],complex);
(%o1)                                     done
(%i2) norm(ket([a,b,c]));
(%o2)          sqrt(c conjugate(c) + b conjugate(b) + a conjugate(a))
(%i3) norm(ket(a,b,c));
(%o3)          norm(|a, b, c>)
```

magsqr (c) [Function]

magsqr returns $\text{conjugate}(c)*c$, the magnitude squared of a complex number.

```
(%i1) declare([a,b,c,d],complex);
(%o1)                                     done
(%i2) A:braket(ket([a,b]),ket([c,d]));
(%o2)          conjugate(b) d + conjugate(a) c
(%i3) P:magsqr(A);
(%o3) (conjugate(b) d + conjugate(a) c) (b conjugate(d) + a conjugate(c))■
```

1.2.1 Handling general kets and bras

General kets and bras are, as discussed, created without using a list when giving the arguments. The following examples show how general kets and bras can be manipulated.

```
(%i1) ket(a)+ket(b);
(%o1)          |b> + |a>
(%i2) braket(bra(a),ket(b));
(%o2)          kron_delta(a, b)
(%i3) braket(bra(a)+bra(c),ket(b));
(%o3)          kron_delta(b, c) + kron_delta(a, b)
```

1.2.2 Spin-1/2 state kets and associated operators

Spin-1/2 particles are characterized by a simple 2-dimensional Hilbert space of states. It is spanned by two vectors. In the z -basis these vectors are $\{z_p, z_m\}$, and the basis kets in the z -basis are $\{x_p, x_m\}$ and $\{y_p, y_m\}$ respectively.

zp [Function]
Return the $|z+\rangle$ ket in the z -basis.

zm [Function]
Return the $|z-\rangle$ ket in the z -basis.

xp [Function]
Return the $|x+\rangle$ ket in the z -basis.

xm [Function]
Return the $|x-\rangle$ ket in the z -basis.

yp [Function]
Return the $|y+\rangle$ ket in the z -basis.

ym [Function]
Return the $|y-\rangle$ ket in the z -basis.

```
(%i1) zp;
                                [ 1 ]
(%o1)                                [  ]
                                [ 0 ]

(%i2) zm;
                                [ 0 ]
(%o2)                                [  ]
                                [ 1 ]

(%i1) yp;
                                [ 1 ]
                                [ ----- ]
                                [ sqrt(2) ]
(%o1)                                [  ]
                                [ %i ]
                                [ ----- ]
                                [ sqrt(2) ]

(%i2) ym;
                                [ 1 ]
                                [ ----- ]
                                [ sqrt(2) ]
(%o2)                                [  ]
                                [ %i ]
                                [ - ----- ]
                                [ sqrt(2) ]
```

```
(%i1) braket(xp,zp);
```

```
(%o1)
          1
        -----
        sqrt(2)
```

Switching bases is done in the following example where a z-basis ket is constructed and the x-basis ket is computed.

```
(%i1) declare([a,b],complex);
```

```
(%o1)
done
```

```
(%i2) psi:ket([a,b]);
```

```
(%o2)
      [ a ]
      [   ]
      [ b ]
```

```
(%i3) psi_x: 'xp*braket(xp,psi)+'xm*braket(xm,psi);
```

```
(%o3)
      b      a      a      b
  (----- + -----) xp + (----- - -----) xm
      sqrt(2)  sqrt(2)  sqrt(2)  sqrt(2)
```

1.2.3 Pauli matrices and Sz, Sx, Sy operators

sigmax [Function]
Returns the Pauli x matrix.

sigmay [Function]
Returns the Pauli y matrix.

sigmaz [Function]
Returns the Pauli z matrix.

Sx [Function]
Returns the spin-1/2 *Sx* matrix.

Sy [Function]
Returns the spin-1/2 *Sy* matrix.

Sz [Function]
Returns the spin-1/2 *Sz* matrix.

```
(%i1) sigmay;
```

```
(%o1)
      [ 0  - %i ]
      [         ]
      [ %i   0  ]
```

```
(%i2) Sy;
```

```
(%o2)
      [          %i hbar ]
      [ 0  - ----- ]
      [          2      ]
      [         ]
      [ %i hbar         ]
      [ -----  0      ]
      [ 2              ]
```

commutator (X,Y) [Function]

Given two operators X and Y, return the commutator $X \cdot Y - Y \cdot X$.

(%i1) commutator(Sx,Sy);

(%o1)

$$\begin{bmatrix} 0 & \frac{i\hbar}{2} \\ \frac{i\hbar}{2} & 0 \end{bmatrix}$$

1.2.4 SX, SY, SZ operators for any spin

SX (s) [Function]

SX(s) for spin **s** returns the matrix representation of the spin operator **Sx**. Shortcuts for spin-1/2 are **Sx,Sy,Sz**, and for spin-1 are **Sx1,Sy1,Sz1**.

SY (s) [Function]

SY(s) for spin **s** returns the matrix representation of the spin operator **Sy**. Shortcuts for spin-1/2 are **Sx,Sy,Sz**, and for spin-1 are **Sx1,Sy1,Sz1**.

SZ (s) [Function]

SZ(s) for spin **s** returns the matrix representation of the spin operator **Sz**. Shortcuts for spin-1/2 are **Sx,Sy,Sz**, and for spin-1 are **Sx1,Sy1,Sz1**.

Example:

(%i1) SY(1/2);

(%o1)

$$\begin{bmatrix} 0 & \frac{i\hbar}{2} \\ \frac{i\hbar}{2} & 0 \end{bmatrix}$$

(%i2) SX(1);

(%o2)

$$\begin{bmatrix} \frac{\hbar}{\sqrt{2}} & 0 \\ 0 & -\frac{\hbar}{\sqrt{2}} \end{bmatrix}$$

1.2.5 Expectation value and variance

`expect (0,psi)` [Function]

Computes the quantum mechanical expectation value of the operator `0` in state `psi`, $\langle \text{psi} | 0 | \text{psi} \rangle$.

```
(%i1) ev(expect(Sy,xp+ym),ratsimp);
(%o1) - hbar
```

`qm_variance (0,psi)` [Function]

Computes the quantum mechanical variance of the operator `0` in state `psi`, $\sqrt{\langle \text{psi} | 0^2 | \text{psi} \rangle - \langle \text{psi} | 0 | \text{psi} \rangle^2}$.

```
(%i1) ev(qm_variance(Sy,xp+ym),ratsimp);
                                %i hbar
(%o1) -----
                                2
```

1.2.6 Angular momentum representation of kets and bras

To create kets and bras in the $|j,m\rangle$ representation you can use the following functions.

`jm_ket (j,m)` [Function]

`jm_ket` creates the ket $|j,m\rangle$ for total spin j and z-component m .

`jm_bra (j,m)` [Function]

`jm_bra` creates the bra $\langle j,m|$ for total spin j and z-component m .

```
(%i1) jm_bra(3/2,1/2);
                                3  1
(%o1)          jm_bra(-, -)
                                2  2

(%i2) jm_bra([3/2,1/2]);
                                [ 3  1 ]
(%o2)          [jmbra, [ -  - ]]
                                [ 2  2 ]
```

`jm_ketp (jmket)` [Function]

`jm_ketp` checks to see that the ket has the 'jmket' marker.

```
(%i1) jm_ketp(jm_ket(j,m));
(%o1) false
(%i2) jm_ketp(jm_ket([j,m]));
(%o2) true
```

`jm_brap (jmbra)` [Function]

`jm_brap` checks to see that the bra has the 'jmbra' marker.

`jm_check (j,m)` [Function]

`jm_check` checks to see that m is one of $\{-j, \dots, +j\}$.

```
(%i1) jm_check(3/2,1/2);
(%o1) true
```

jm_braket (*jm_bra, jm_ket*)

[Function]

jm_braket takes the inner product of the jm-kets.

```
(%i1) K:jm_ket(j1,m1);
(%o1)                                     jm_ket(j1, m1)
(%i2) B:jm_bra(j2,m2);
(%o2)                                     jm_bra(j2, m2)
(%i3) jm_braket(B,K);
(%o3)      kron_delta(j1, j2) kron_delta(m1, m2)
(%i4) B:jm_bra(j1,m1);
(%o4)                                     jm_bra(j1, m1)
(%i5) jm_braket(B,K);
(%o5)                                     1
(%i6) K:jm_ket([j1,m1]);
(%o6)      [jmket, [ j1  m1 ]]
(%i7) B:jm_bra([j2,m2]);
(%o7)      [jmbra, [ j2  m2 ]]
(%i8) jm_braket(B,K);
(%o8)                                     0
(%i9) jm_braket(jm_bra(j1,m1)+jm_bra(j3,m3),jm_ket(j2,m2));
(%o9) kron_delta(j2, j3) kron_delta(m2, m3)
      + kron_delta(j1, j2) kron_delta(m1, m2)■
```

JP (*jm_ket*)

[Function]

JP is the J_+ operator. It takes a jmket jm_ket(j,m) and returns $\sqrt{j*(j+1)-m*(m+1)}*\hbar*jm_ket(j,m+1)$.

JM (*jm_ket*)

[Function]

JM is the J_- operator. It takes a jmket jm_ket(j,m) and returns $\sqrt{j*(j+1)-m*(m-1)}*\hbar*jm_ket(j,m-1)$.

Jsqr (*jm_ket*)

[Function]

Jsqr is the J^2 operator. It takes a jmket jm_ket(j,m) and returns $(j*(j+1)*\hbar^2*jm_ket(j,m))$.

Jz (*jm_ket*)

[Function]

Jz is the J_z operator. It takes a jmket jm_ket(j,m) and returns $m*\hbar*jm_ket(j,m)$.

These functions are illustrated below.

```

(%i1) k:jm_ket([j,m]);
(%o1) [jmket, [ j m ]]
(%i2) JP(k);
(%o2) hbar jm_ket(j, m + 1) sqrt(j (j + 1) - m (m + 1))
(%i3) JM(k);
(%o3) hbar jm_ket(j, m - 1) sqrt(j (j + 1) - (m - 1) m)
(%i4) Jsqr(k);
(%o4) hbar j (j + 1) jm_ket(j, m)
(%i5) Jz(k);
(%o5) hbar jm_ket(j, m) m

```

1.2.7 Angular momentum and ladder operators

SP (s) [Function]
 SP is the raising ladder operator S_+ for spin s .

SM (s) [Function]
 SM is the raising ladder operator S_- for spin s .

Examples of the ladder operators:

```

(%i1) SP(1);
(%o1) [ 0 sqrt(2) hbar 0 ]
      [ 0 0 sqrt(2) hbar ]
      [ 0 0 0 0 ]
(%i2) SM(1);
(%o2) [ 0 0 0 0 ]
      [ sqrt(2) hbar 0 0 ]
      [ 0 sqrt(2) hbar 0 ]

```

1.3 Rotation operators

RX (s,t) [Function]
 RX(s) for spin s returns the matrix representation of the rotation operator R_x for rotation through angle t .

RY (s,t) [Function]
 RY(s) for spin s returns the matrix representation of the rotation operator R_y for rotation through angle t .

RZ (s,t) [Function]
 RZ(s) for spin s returns the matrix representation of the rotation operator R_z for rotation through angle t .

```
(%i1) RZ(1/2,t);
Proviso: assuming 64*t # 0
```

$$\begin{bmatrix}
e^{i t/2} & 0 \\
0 & e^{-i t/2}
\end{bmatrix}$$

```
(%o1)
```

1.4 Time-evolution operator

UU (H,t) [Function]

UU(H,t) is the time evolution operator for Hamiltonian H. It is defined as the matrix exponential `matrixexp(-%i*H*t/hbar)`.

```
(%i1) UU(w*Sy,t);
Proviso: assuming 64*t*w # 0
```

$$\begin{bmatrix}
\cos\left(\frac{t w}{2}\right) & -i \sin\left(\frac{t w}{2}\right) \\
i \sin\left(\frac{t w}{2}\right) & \cos\left(\frac{t w}{2}\right)
\end{bmatrix}$$

```
(%o1)
```

1.5 Tensor products

Tensor products are represented as lists in Maxima. The ket tensor product $|z+,z+\rangle$ is represented as `[tpket,zp,zp]`, and the bra tensor product $\langle a,b|$ is represented as `[tpbra,a,b]` for kets **a** and **b**. The list labels **tpket** and **tpbra** ensure calculations are performed with the correct kind of objects.

ketprod (k₁, k₂, ...) [Function]

ketprod produces a tensor product of kets **k_i**. All of the elements must pass the **ketp** predicate test to be accepted.

braprod (b₁, b₂, ...) [Function]

braprod produces a tensor product of bras **b_i**. All of the elements must pass the **brap** predicate test to be accepted.

braketprod (B,K) [Function]

braketprod takes the inner product of the tensor products **B** and **K**. The tensor products must be of the same length (number of kets must equal the number of bras).

Examples below show how to create tensor products and take the bracket of tensor products.

```

(%i1) ketprod(zp,zm);
(%o1)

$$\text{ketprod}\left(\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix}\right)$$

(%i2) ketprod('zp,'zm);
(%o2) ketprod(zp, zm)
(%i1) kill(a,b,c,d);
(%o1) done
(%i2) declare([a,b,c,d],complex);
(%o2) done
(%i3) braprod(bra([a,b]),bra([c,d]));
(%o3) braprod([ a b ], [ c d ])
(%i4) braprod(dagger(zp),bra([c,d]));
(%o4) braprod([ 1 0 ], [ c d ])
(%i1) K:ketprod(zp,zm);
(%o1)

$$\text{ketprod}\left(\begin{bmatrix} 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \end{bmatrix}\right)$$

(%i2) zpb:dagger(zp);
(%o2) [ 1 0 ]
(%i3) zmb:dagger(zm);
(%o3) [ 0 1 ]
(%i4) B:braprod(zpb,zmb);
(%o4) braprod([ 1 0 ], [ 0 1 ])
(%i5) brakprod(K,B);
(%o5) false
(%i6) brakprod(B,K);
(%o6) false

```

Appendix A Function and Variable index

A

autobra	4
autoket	4

B

bra	3
braket	5
braketprod	12
brap	4
braprod	12

C

commutator	8
------------------	---

D

dagger	4
--------------	---

E

expect	9
--------------	---

J

jm_bra	9
jm_braket	10
jm_brap	9
jm_check	9
jm_ket	9
jm_ketp	9
JM	10
JP	10
Jsqr	10
Jz	10

K

ket	2
ketp	3
ketprod	12

M

magsqr	5
--------------	---

N

norm	5
------------	---

Q

qm_variance	9
-------------------	---

R

RX	11
RY	11
RZ	11

S

sigmax	7
sigmay	7
sigmaz	7
SM	11
SP	11
Sx	7
SX	8
Sy	7
SY	8
Sz	7
SZ	8

U

UU	12
----------	----

X

xm	6
xp	6

Y

ym	6
yp	6

Z

zm	6
zp	6

hbar 2