

1.1 Introduction to package `qm`

The `qm` package provides functions and standard definitions to solve quantum mechanics problems in a finite dimensional Hilbert space. For example, one can calculate the outcome of Stern-Gerlach experiments using the built-in definition of the S_x , S_y , and S_z operators for arbitrary spin, e.g. $s=\{1/2, 1, 3/2, \dots\}$. For spin-1/2 the standard basis states in the x , y , and z -basis are available as $\{x_p, x_m\}$, $\{y_p, y_m\}$, and $\{z_p, z_m\}$. One can create general ket vectors with arbitrary but finite dimension and perform standard computations such as expectation value, variance, etc. The angular momentum $|j, m\rangle$ representation of kets is also available. It is also possible to create tensor product states for multiparticle systems and to perform calculations on those systems.

Kets and bras are represented by column and row vectors, respectively. For spin-1/2 particles, for example, the bra vector

$$\langle \text{psi} | = a \langle z+ | + b \langle z- |$$

is represented by the row vector $[a \ b]$, where the basis vectors $\langle z+ | = [1 \ 0]$ and $\langle z- | = [0 \ 1]$.

The `qm` package was written by Eric Majzoub, University of Missouri. (Email: majzoub@at-umsystem.edu) The package is loaded with: `load(qm)`;

1.2 Functions and Variables for `qm`

hbar [Variable]

Planck's constant divided by 2π . `hbar` is not given a floating point value, but is declared to be a real number greater than zero.

ket ($[c_1, c_2, \dots]$) [Function]

`ket` creates a *column* vector of arbitrary finite dimension. The entries c_i can be any Maxima expression. The user must **declare** any relevant constants to be complex. For a matrix representation the elements must be entered as a list in $[\dots]$ square brackets. If no list is entered the ket is represented as a general ket, `ket(a)` will return $|a\rangle$.

```
(%i4) kill(a);
(%o4)                                     done
(%i5) ket(a);
(%o5)                                     |a>
(%i6) declare([c1,c2],complex);
(%o6)                                     done
(%i7) ket([c1,c2]);
                                     [ c1 ]
(%o7)                                     [   ]
                                     [ c2 ]

(%i8) facts();
(%o8) [kind(hbar, real), hbar > 0, kind(c1, complex), kind(c2, complex)]
```

bra ($[c_1, c_2, \dots]$) [Function]

`bra` creates a *row* vector of arbitrary finite dimension. The entries c_i can be any Maxima expression. The user must **declare** any relevant constants to be complex.

For a matrix representation the elements must be entered as a list in [...] square braces. If no list is entered the bra is represented as a general bra, `bra(a)` will return `<a|`.

```
(%i4) kill(c1,c2);
(%o4)                                     done
(%i5) bra(c1,c2);
(%o5)                                     <c1, c2|
(%i6) bra([c1,c2]);
(%o6)                                     [ c1  c2 ]
(%i7) facts();
(%o7)                                     [kind(hbar, real), hbar > 0]
```

ketp (*vector*) [Function]

ketp is a predicate function that checks if its input is a ket, in which case it returns **true**, else it returns **false**. **ketp** only returns **true** for the matrix representation of a ket.

```
(%i4) kill(a,b,k);
(%o4)                                     done
(%i5) k:ket(a,b);
(%o5)                                     |a, b>
(%i6) ketp(k);
(%o6)                                     false
(%i7) k:ket([a,b]);
(%o7)                                     [ a ]
                                     [  ]
                                     [ b ]
(%i8) ketp(k);
(%o8)                                     true
```

brap (*vector*) [Function]

brap is a predicate function that checks if its input is a bra, in which case it returns **true**, else it returns **false**. **brap** only returns **true** for the matrix representation of a bra.

```
(%i4) b:bra([a,b]);
(%o4)                                     [ a  b ]
(%i5) brap(b);
(%o5)                                     true
```

dagger (*vector*) [Function]

dagger is the quantum mechanical *dagger* function and returns the **conjugate transpose** of its input.

```
(%i4) dagger(bra([%i,2]));
(%o4)                                     [ - %i ]
                                     [      ]
                                     [  2    ]
```

braket (psi,phi) [Function]

Given two kets **psi** and **phi**, **braket** returns the quantum mechanical bracket $\langle \text{psi} | \text{phi} \rangle$. The vector **psi** may be input as either a **ket** or **bra**. If it is a **ket** it will be turned into a **bra** with the **dagger** function before the inner product is taken. The vector **phi** must always be a **ket**.

```
(%i4) declare([a,b,c],complex);
(%o4)
done
(%i5) braket(ket([a,b,c]),ket([a,b,c]));
(%o5) c conjugate(c) + b conjugate(b) + a conjugate(a)
```

norm (psi) [Function]

Given a **ket** or **bra** **psi**, **norm** returns the square root of the quantum mechanical bracket $\langle \text{psi} | \text{psi} \rangle$. The vector **psi** must always be a **ket**, otherwise the function will return **false**.

```
(%i4) declare([a,b,c],complex);
(%o4)
done
(%i5) norm(ket([a,b,c]));
(%o5) sqrt(c conjugate(c) + b conjugate(b) + a conjugate(a))
(%i6) norm(ket(a,b,c));
(%o6) sqrt(braket(|a, b, c>, |a, b, c>))
```

magsqr (c) [Function]

magsqr returns $\text{conjugate}(c)*c$, the magnitude squared of a complex number.

```
(%i4) declare([a,b,c,d],complex);
(%o4)
done
(%i5) A:braket(ket([a,b]),ket([c,d]));
(%o5) conjugate(b) d + conjugate(a) c
(%i6) P:magsqr(A);
(%o6) (conjugate(b) d + conjugate(a) c) (b conjugate(d) + a conjugate(c))■
```

1.2.1 Spin-1/2 state kets and associated operators

Spin-1/2 particles are characterized by a simple 2-dimensional Hilbert space of states. It is spanned by two vectors. In the z-basis these vectors are $\{\text{zp}, \text{zm}\}$, and the basis kets in the z-basis are $\{\text{xp}, \text{xm}\}$ and $\{\text{yp}, \text{ym}\}$ respectively.

zp [Function]

Return the $|z+\rangle$ ket in the z-basis.

zm [Function]

Return the $|z-\rangle$ ket in the z-basis.

xp [Function]

Return the $|x+\rangle$ ket in the z-basis.

xm [Function]

Return the $|x-\rangle$ ket in the z-basis.

yp [Function]

Return the $|y+\rangle$ ket in the z-basis.

ym [Function]

Return the $|y-\rangle$ ket in the z-basis.

```
(%i4) zp;
(%o4)      [ 1 ]
           [   ]
           [ 0 ]

(%i5) zm;
(%o5)      [ 0 ]
           [   ]
           [ 1 ]

(%i4) yp;
(%o4)      [ 1 ]
           [ ----- ]
           [ sqrt(2) ]
           [   ]
           [ %i ]
           [ ----- ]
           [ sqrt(2) ]

(%i5) ym;
(%o5)      [ 1 ]
           [ ----- ]
           [ sqrt(2) ]
           [   ]
           [ %i ]
           [ - ----- ]
           [ sqrt(2) ]

(%i4) brakel(xp,zp);
(%o4)      1
           -----
           sqrt(2)
```

Switching bases is done in the following example where a z-basis ket is constructed and the x-basis ket is computed.

```
(%i4) declare([a,b],complex);
(%o4)      done
(%i5) psi:ket([a,b]);
(%o5)      [ a ]
           [   ]
           [ b ]

(%i6) psi_x:'xp*braket(xp,psi)+'xm*braket(xm,psi);
(%o6)      (----- + -----) xp + (----- - -----) xm
           b      a      a      b
          sqrt(2) sqrt(2) sqrt(2) sqrt(2)
```

1.2.2 Pauli matrices and Sz, Sx, Sy operators

sigmax [Function]
Returns the Pauli x matrix.

sigmay [Function]
Returns the Pauli y matrix.

sigmaz [Function]
Returns the Pauli z matrix.

Sx [Function]
Returns the spin-1/2 *Sx* matrix.

Sy [Function]
Returns the spin-1/2 *Sy* matrix.

Sz [Function]
Returns the spin-1/2 *Sz* matrix.

(%i4) sigmay;

(%o4)

$$\begin{bmatrix} 0 & -i \\ i & 0 \end{bmatrix}$$

(%i5) Sy;

(%o5)

$$\begin{bmatrix} 0 & \frac{i \hbar}{2} \\ \frac{i \hbar}{2} & 0 \end{bmatrix}$$

commutator (X,Y) [Function]
Given two operators X and Y, return the commutator $X \cdot Y - Y \cdot X$.

(%i4) commutator(Sx,Sy);

(%o4)

$$\begin{bmatrix} \frac{\hbar}{2} & 0 \\ 0 & -\frac{\hbar}{2} \end{bmatrix}$$

1.2.3 SX, SY, SZ operators for any spin

SX (s) [Function]
SX(s) for spin *s* returns the matrix representation of the spin operator *Sx*. Shortcuts for spin-1/2 are *Sx*,*Sy*,*Sz*, and for spin-1 are *Sx1*,*Sy1*,*Sz1*.

SY (s) [Function]
 SY(s) for spin **s** returns the matrix representation of the spin operator **Sy**. Shortcuts for spin-1/2 are **Sx,Sy,Sz**, and for spin-1 are **Sx1,Sy1,Sz1**.

SZ (s) [Function]
 SZ(s) for spin **s** returns the matrix representation of the spin operator **Sz**. Shortcuts for spin-1/2 are **Sx,Sy,Sz**, and for spin-1 are **Sx1,Sy1,Sz1**.

Example:

(%i4) SY(1/2);

(%o4)

$$\begin{bmatrix} 0 & -\frac{i\hbar}{2} \\ \frac{i\hbar}{2} & 0 \end{bmatrix}$$

(%i5) SX(1);

(%o5)

$$\begin{bmatrix} \hbar & 0 \\ 0 & -\hbar \end{bmatrix}$$

1.2.4 Expectation value and variance

expect (O,psi) [Function]
 Computes the quantum mechanical expectation value of the operator **O** in state **psi**, $\langle \text{psi} | O | \text{psi} \rangle$.

(%i4) ev(expect(Sy,xp+ym),ratsimp);
 (%o4) $-\hbar$

qm_variance (O,psi) [Function]
 Computes the quantum mechanical variance of the operator **O** in state **psi**, $\sqrt{\langle \text{psi} | O^2 | \text{psi} \rangle - \langle \text{psi} | O | \text{psi} \rangle^2}$.

(%i4) ev(qm_variance(Sy,xp+ym),ratsimp);
 (%o4) $\frac{\hbar}{2}$

1.2.5 Angular momentum representation of kets and bras

To create kets and bras in the $|j,m\rangle$ representation you can use the following functions.

`jm_ket (j,m)` [Function]
`jm_ket` creates the ket $|j,m\rangle$ for total spin j and z-component m .

`jm_bra (j,m)` [Function]
`jm_bra` creates the bra $\langle j,m|$ for total spin j and z-component m .

```
(%i4) jm_bra(3/2,1/2);
(%o4) [jm_bra, [ 3 1 ]
          [ - - ]
          [ 2 2 ]]
```

`jm_ketp (jmket)` [Function]
`jm_ketp` checks to see that the ket has the 'jmket' marker.

`jm_brap (jmbra)` [Function]
`jm_brap` checks to see that the bra has the 'jmbra' marker.

`jm_check (j,m)` [Function]
`jm_check` checks to see that m is one of $\{-j, \dots, +j\}$.

`jm_braket (jmbra,jmket)` [Function]
`jm_braket` takes the inner product of the jm-kets.

```
(%i4) K:jm_ket(j1,m1);
(%o4) [jmket, [ j1 m1 ]]
(%i5) B:jm_bra(j2,m2);
(%o5) [jmbra, [ j2 m2 ]]
(%i6) jm_braket(B,K);
(%o6) 0
(%i7) B:jm_bra(j1,m1);
(%o7) [jmbra, [ j1 m1 ]]
(%i8) jm_braket(B,K);
(%o8) 1
```

1.2.6 Angular momentum and ladder operators

`SP (s)` [Function]
`SP` is the raising ladder operator S_+ for spin s .

`SM (s)` [Function]
`SM` is the raising ladder operator S_- for spin s .

Examples of the ladder operators:

```
(%i4) SP(1);
[ 0  sqrt(2) hbar      0      ]
[                                ]
(%o4) [ 0      0      sqrt(2) hbar ]
[                                ]
[ 0      0      0      ]

(%i5) SM(1);
[      0      0      0 ]
[                                ]
(%o5) [ sqrt(2) hbar      0      0 ]
[                                ]
[      0      sqrt(2) hbar  0 ]
```

1.3 Rotation operators

RX (s,t) [Function]
 RX(s) for spin **s** returns the matrix representation of the rotation operator **R_x** for rotation through angle **t**.

RY (s,t) [Function]
 RY(s) for spin **s** returns the matrix representation of the rotation operator **R_y** for rotation through angle **t**.

RZ (s,t) [Function]
 RZ(s) for spin **s** returns the matrix representation of the rotation operator **R_z** for rotation through angle **t**.

```
(%i4) RZ(1/2,t);
Proviso: assuming 64*t # 0
[      %i t      ]
[  -  ----      ]
[      2          ]
[ %e      0      ]
(%o4) [                                ]
[      %i t      ]
[  ----          ]
[      2          ]
[  0      %e      ]
```

1.4 Time-evolution operator

UU (H,t) [Function]
 UU(H,t) is the time evolution operator for Hamiltonian **H**. It is defined as the matrix exponential `matrixexp(-%i*H*t/hbar)`.


```
(%i4) UU(w*Sy,t);
Proviso: assuming 64*t*w # 0
[      t w      t w ]
[ cos(---) - sin(---) ]
[      2      2 ]
(%o4) [ ]
[      t w      t w ]
[ sin(---)  cos(---) ]
[      2      2 ]
```

1.5 Tensor products

Tensor products are represented as lists in Maxima. The ket tensor product $|z+,z+\rangle$ is represented as `[tpket,zp,zp]`, and the bra tensor product $\langle a,b|$ is represented as `[tpbra,a,b]` for kets `a` and `b`. The list labels `tpket` and `tpbra` ensure calculations are performed with the correct kind of objects.

ketprod (k_1, k_2, \dots) [Function]
ketprod produces a tensor product of kets k_i . All of the elements must pass the **ketp** predicate test to be accepted.

braprod (b_1, b_2, \dots) [Function]
braprod produces a tensor product of bras b_i . All of the elements must pass the **brap** predicate test to be accepted.

braketprod (B,K) [Function]
braketprod takes the inner product of the tensor products B and K. The tensor products must be of the same length (number of kets must equal the number of bras).

Examples below show how to create tensor products and take the bracket of tensor products.

```
(%i4) ketprod(zp,zm);
[ 1 ] [ 0 ]
(%o4) [tpket, [[ ], [ ]]]
[ 0 ] [ 1 ]

(%i5) ketprod('zp,'zm);
all elements must be kets

(%o5) done
(%i4) kill(a,b,c,d);
(%o4) done
(%i5) declare([a,b,c,d],complex);
(%o5) done
(%i6) braprod(bra([a,b]),bra([c,d]));
(%o6) [tpbra, [[ a b ], [ c d ]]]
(%i7) braprod(dagger(zp),bra([c,d]));
(%o7) [tpbra, [[ 1 0 ], [ c d ]]]
```

```

(%i4) K:ketprod(zp,zm);
                                [ 1 ] [ 0 ]
(%o4)      [tpket, [[  ], [  ]]]
                                [ 0 ] [ 1 ]

(%i5) zpb:dagger(zp);
(%o5)      [ 1 0 ]

(%i6) zmb:dagger(zm);
(%o6)      [ 0 1 ]

(%i7) B:braprod(zpb,zmb);
(%o7)      [tpbra, [[ 1 0 ], [ 0 1 ]]]

(%i8) braketprod(K,B);
(%o8)      false

(%i9) braketprod(B,K);
(%o9)      1

```

Appendix A Function and Variable index

B

| | |
|-----------------|---|
| bra..... | 1 |
| braket..... | 3 |
| braketprod..... | 9 |
| brap..... | 2 |
| braprod..... | 9 |

C

| | |
|-----------------|---|
| commutator..... | 5 |
|-----------------|---|

D

| | |
|-------------|---|
| dagger..... | 2 |
|-------------|---|

E

| | |
|-------------|---|
| expect..... | 6 |
|-------------|---|

J

| | |
|----------------|---|
| jm_bra..... | 7 |
| jm_braket..... | 7 |
| jm_brap..... | 7 |
| jm_check..... | 7 |
| jm_ket..... | 7 |
| jm_ketp..... | 7 |

K

| | |
|--------------|---|
| ket..... | 1 |
| ketp..... | 2 |
| ketprod..... | 9 |

M

| | |
|-------------|---|
| magsqr..... | 3 |
|-------------|---|

N

| | |
|-----------|---|
| norm..... | 3 |
|-----------|---|

Q

| | |
|------------------|---|
| qm_variance..... | 6 |
|------------------|---|

R

| | |
|---------|---|
| RX..... | 8 |
| RY..... | 8 |
| RZ..... | 8 |

S

| | |
|-------------|---|
| sigmax..... | 5 |
| sigmay..... | 5 |
| sigmaz..... | 5 |
| SM..... | 7 |
| SP..... | 7 |
| Sx..... | 5 |
| SX..... | 5 |
| Sy..... | 5 |
| SY..... | 6 |
| Sz..... | 5 |
| SZ..... | 6 |

U

| | |
|---------|---|
| UU..... | 8 |
|---------|---|

X

| | |
|---------|---|
| xm..... | 3 |
| xp..... | 3 |

Y

| | |
|---------|---|
| ym..... | 4 |
| yp..... | 4 |

Z

| | |
|---------|---|
| zm..... | 3 |
| zp..... | 3 |

hbar 1